

NAG Library Routine Document

G05SSF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G05SSF generates a vector of pseudorandom numbers from a two parameter Weibull distribution with shape parameter a and scale parameter b .

2 Specification

```
SUBROUTINE G05SSF(N, A, B, STATE, X, IFAIL)
INTEGER          N, STATE(lstate), IFAIL
double precision A, B, X(N)
```

3 Description

The distribution has PDF (probability density function)

$$f(x) = \frac{a}{b} x^{a-1} e^{-x^a/b} \quad \text{if } x > 0,$$

$$f(x) = 0 \quad \text{otherwise.}$$

G05SSF returns the value $(-b \ln y)^{1/a}$, where y is a pseudorandom number from a uniform distribution over $(0, 1]$.

One of the initialization routines G05KFF (for a repeatable sequence if computed sequentially) or G05KGF (for a non-repeatable sequence) must be called prior to the first call to G05SSF.

4 References

Kendall M G and Stuart A (1969) *The Advanced Theory of Statistics (Volume 1)* (3rd Edition) Griffin
 Knuth D E (1981) *The Art of Computer Programming (Volume 2)* (2nd Edition) Addison–Wesley

5 Parameters

- | | | |
|----|--|--------------|
| 1: | N – INTEGER | <i>Input</i> |
| | <i>On entry:</i> n , the number of pseudorandom numbers to be generated. | |
| | <i>Constraint:</i> $N \geq 0$. | |
| 2: | A – double precision | <i>Input</i> |
| | <i>On entry:</i> a , the shape parameter of the distribution. | |
| | <i>Constraint:</i> $A > 0.0$. | |
| 3: | B – double precision | <i>Input</i> |
| | <i>On entry:</i> b , the scale parameter of the distribution. | |
| | <i>Constraint:</i> $B > 0.0$. | |

- 4: STATE(*lstate*) – INTEGER array *Communication Array*
Note: *lstate* = LSTATE, the dimension of STATE as supplied to the initialization routine G05KFF or G05KGF.
On entry: contains information on the selected base generator and its current state.
On exit: contains updated information on the state of the generator.
- 5: X(N) – **double precision** array *Output*
On exit: the *n* pseudorandom numbers from the specified Weibull distribution.
- 6: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, –1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.
On exit: IFAIL = 0 unless the routine detects an error (see Section 6).
 For environments where it might be inappropriate to halt program execution when an error is detected, the value –1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value –1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or –1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, $N < 0$.

IFAIL = 2

On entry, $A \leq 0.0$.

IFAIL = 3

On entry, $B \leq 0.0$.

IFAIL = 4

On entry, STATE vector was not initialized or has been corrupted.

7 Accuracy

Not applicable.

8 Further Comments

None.

9 Example

This example prints the first five pseudorandom numbers from a Weibull distribution with shape parameter 1.0 and scale parameter 2.0, generated by a single call to G05SSF, after initialization by G05KFF.

9.1 Program Text

```

*      G05SSF Example Program Text
*      Mark 22 Release. NAG Copyright 2006.
*      .. Parameters ..
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
      INTEGER          MSTATE, MSEED, N
      PARAMETER        (MSTATE=633,MSEED=1,N=5)
*      .. Local Scalars ..
      DOUBLE PRECISION A, B
      INTEGER          GENID, I, IFAIL, LSEED, LSTATE, SUBID
*      .. Local Arrays ..
      DOUBLE PRECISION X(N)
      INTEGER          SEED(MSEED), STATE(MSTATE)
*      .. External Subroutines ..
      EXTERNAL         G05KFF, G05SSF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'G05SSF Example Program Results'
      WRITE (NOUT,*)
*      Set the distribution parameters A and B
      A = 1.0D0
      B = 2.0D0
*      Initialize the seed to a repeatable sequence
      SEED(1) = 1762543
*      GENID and SUBID identify the base generator.
      GENID = 1
      SUBID = 1
*      Initialize the generator to a repeatable sequence
      LSTATE = MSTATE
      LSEED = MSEED
      IFAIL = 1
      CALL G05KFF(GENID,SUBID,SEED,LSEED,STATE,LSTATE,IFAIL)
      IF (IFAIL.NE.0) THEN
         WRITE (NOUT,99998) IFAIL
         GO TO 20
      END IF
*      Generate the variates
      IFAIL = 1
      CALL G05SSF(N,A,B,STATE,X,IFAIL)
      IF (IFAIL.NE.0) THEN
         WRITE (NOUT,99997) IFAIL
         GO TO 20
      END IF
*      Display the variates
      WRITE (NOUT,99999) (X(I),I=1,N)
*
      20 CONTINUE
*
99999  FORMAT (1X,F10.4)
99998  FORMAT (1X,' ** G05KFF returned with IFAIL = ',I5)
99997  FORMAT (1X,' ** G05SSF returned with IFAIL = ',I5)
      END

```

9.2 Program Data

None.

9.3 Program Results

G05SSF Example Program Results

```

0.9039
4.4796
0.5860
0.4506
4.5154

```